
The protocol

1. A −→ S : A,B,RA

2. S −→ A : {RA, B,KAB, {KAB, A}KBS
}KAS

3. A −→ B : {KAB, A}KBS

4. B −→ A : {RB}KAB

5. A −→ B : {RB − 1}KAB

An APA model for the symmetric Needham-Schroeder protocol and a detailed
explanation of that model is given in [SOR02]. Here a translation of a sligthly
modified version of that model into the syntax of the SH verification tool is shown.

The reachability graph for a simple protocol run computed by the tool is ap-
pended.

As a basis for the implementation the APA model as specified in the appendix
of [SOR02] is used.

Corresponding syntax of the APA model and SH verification tool [Fra02b,
Fra02a, ORR00] are shown side by side.

The complete specification in SH verification tool syntax is included in the
appendix.

A short tour of SH Verification Tool syntax for APA state transition pattern
notation:

Some examples used in the Needham-Schroeder model and their SH verifica-
tion tool counterpart:

(start, B) ∈ StateA [’start’,B] ? A State

(start, B)←↩ StateA tool syntax: [’start’,B] << A State

(B,RA, S) ↪→ StateA [B,RA,S] >> A State

Example 1: A simple preamble for the SH verification tool and the computed
reachability graph:

A global statemultiset is defined and initialised:
defset nat0_seq = seq(nat_0);
def_state multiset: nat0_seq := 2.3.3;

A role role is defined and bound to’instance’ :
def_role role;
def_pattern_bind role := ’instance’ ;

A transition pattern for that role just takes out one element from the global statemultiset :
def_trans_pattern role pattern

(x)
x << multiset;

Figure 1: Reachability graph of example 1

Example 2: An example using a global statenetwork and two instances of
different roles one putting some information from local staterole1 my set into
global statenetwork and the other taking the data out ofnetwork into the local
staterole2 my set .

defset nat0_seq = seq(nat_0);
def_state network: nat0_seq := ::;

def_role role1;
def_role role2;
def_pattern_bind role1 := ’instance1’ ;
def_pattern_bind role2 := ’instance2’ ;

def_state role1 my_set: nat0_seq := 2.3.3;
def_state role2 my_set: nat0_seq := ::;

def_trans_pattern role1 pattern1
(x)
x << role1_my_set,
network = ::, /* check whether network is empty */
x >> network;

def_trans_pattern role2 pattern1
(y)
y << network,
y >> role2_my_set;

Figure 2 shows the reachability graph computed by the SH Verification Tool
for this example.

Figure 2: Reachability graph of example 2

Now the corresponding syntax of the APA model in [SOR02] and SH Verifi-
cation Tool [Fra02b, Fra02a, ORR00] are shown side by side.

APA model (initial state):

StateA := {(B,agent), (S, server), (start,B)}
SymkeysA := {(S, sym, (A,S, sym))}
AsymkeysA := ∅

StateB := {(A,agent), (S, server), (respond,A)}
SymkeysB := {(S, sym, (B,S, sym))}
AsymkeysB := ∅

StateS := {(A,agent), (B,agent)}
SymkeysS := {(A, sym, (A,S, sym)), (B, sym, (B,S, sym))}
AsymkeysS := ∅

Network := ∅

SH Verification Tool:

def_role A;
def_role B;
def_role S;

def_state A State: Messages_seq := [B,’agent’].[S,’server’].[’start’,B];
def_state A Symkeys: Symkeys_seq := (S,’sym’,(A,S,’sym’));
def_state A Asymkeys: Asymkeys_seq := ::;

def_state B State: Messages_seq := [A,’agent’].[S,’server’].[’respond’,A];
def_state B Symkeys: Symkeys_seq := (S,’sym’,(B,S,’sym’));
def_state B Asymkeys: Asymkeys_seq := ::;

def_state S State: Messages_seq := [A,’agent’].[B,’agent’];
def_state S Symkeys: Symkeys_seq := (A,’sym’,(A,S,’sym’)).

(B,’sym’,(B,S,’sym’));
def_state S Asymkeys: Asymkeys_seq := ::;

def_state Network: net_elem_seq := ::;

def_pattern_bind A := ’Alice’;
def_pattern_bind B := ’Bob’;
def_pattern_bind S := ’Server’;

APA model (step 1):
Variables: RA, B, S

(start, B) ∈ StateA
(B, agent) ∈ StateA
(S, server) ∈ StateA
A→
RA ∈ newnonce
(start, B) ←↩ StateA
(B,RA, S) ↪→ StateA
(A, S, (A,B,RA)) ↪→ Network

SH Verification Tool:

def_trans_pattern A step_1
(RA,B,S)
[’start’,B] ? A_State,
[B,’agent’] ? A_State,
[S,’server’] ? A_State,
RA << new_nonce,
tail(RA) >> new_nonce,
[’start’,B] << A_State,
[B,head(RA),S] >> A_State,
(A,S,[A,B,head(RA)]) >> Network;

For the exampleNonces are modeled in the tool by a list ofNonce . The state
new nonce initially contains a multiset (sequence) of one list ofNonces . Here
we use only two nonces 13 and 17 needed for one run of the protocol in step 1
above and step 4.

defset Nonce = 13, 17, 19, 23;
defset Nonces = [Nonce];
defset Nonce_seq = seq(Nonces);
def_state new_nonce: Nonce_seq := [13,17];

In the pattern aboveRA is bound to the list[13,17] , the single element of the
multisetnew nonce , then the tail ofRA (in this case[17] ) is put back into the
multiset and the head ofRA (in this case13) is used in the further lines of the
above pattern.

In a concrete implementation a pseudo-random number generator should be
used to generate nonces.

APA model (step 2):
Variables: X,A,B,M,KAS, KBS, KAB, RA

(X,S,M) ∈ Network
(A,agent) ∈ StateS∧ elem(1,M) = A
(B, agent) ∈ StateS∧ elem(2,M) = B
(A, sym,KAS) ∈ SymkeysS
(B, sym,KBS) ∈ SymkeysS
S→
RA := elem(3,M)
(A,B, sym) ∈ newkey
KAB := (A,B, sym)
(X,S,M) ←↩ Network
(S,A, (encrypt(KAS, (RA, B,KAB,
encrypt(KBS, (KAB, A)))))) ↪→ Network

SH Verification Tool:

def_trans_pattern S step_2
(X,A,B,M,KAS,KBS)
(X,S,M) ? Network,
[A,’agent’] ? S_State,
A = head(M),
[B,’agent’] ? S_State,
B = head(tail(M)),
(A,’sym’,KAS) ? S_Symkeys,
(B,’sym’,KBS) ? S_Symkeys,

/* RA = head(tail(tail(M))),*/
(X,S,M) << Network,
(S,A,encrypt(KAS,[head(tail(tail(M))),B,(A,B,’sym’),

encrypt(KBS,[(A,B,’sym’),A])])) >> Network;

In the tool the abbreviationRAused in the pattern above is currently not used.
Insteadelem(3,M)is inserted directly notated here byhead(tail(tail(M))) .

APA model (step 3):
Variables: X,B, S,M,Chiffretext, KAS, KAB, RA

(X,A,M) ∈ Network
(B,RA, S) ∈ StateA
(S, sym,KAS) ∈ SymkeysA
elem(1, decrypt(KAS,M)) = RA

elem(2, decrypt(KAS,M)) = B
A→
KAB := elem(3, decrypt(KAS,M))
Chiffretext:= elem(4, decrypt(KAS,M))
(X,A,M) ←↩ Network
(B,RA, S) ←↩ StateA
(new session key, B,KAB) ↪→ StateA
(A,B,Chiffretext) ↪→ Network

SH Verification Tool:

def_trans_pattern A step_3
(X,B,S,M,KAS,RA)
(X,A,M) ? Network,

/* [B,RA,S] ? A_State,*/
(S,’sym’,KAS) ? A_Symkeys,
head(decrypt(KAS,M)) = RA,
head(tail(decrypt(KAS,M))) = B,
[B,RA,S] ? A_State,

/* KAB = head(tail(tail(decrypt(KAS,M)))),*/
/* Chiffretext = head(tail(tail(tail(decrypt(KAS,M))))),*/

(X,A,M) << Network,
[B,RA,S] << A_State,
[new_session_key,B,head(tail(tail(decrypt(KAS,M))))] >> A_State,
(A,B,head(tail(tail(tail(decrypt(KAS,M)))))) >> Network;

In the tool the abbreviationsKABandChiffretext used in the pattern above
are currently not used and the corresponding expression shown in the commented
lines is inserted directly.

8



APA model (step 4):
Variables: X,A, S,M,KBS, KAB, RB

(X,B,M) ∈ Network
(respond, A) ∈ StateB
(A,agent) ∈ StateB
(S, server) ∈ StateB
(S, sym,KBS) ∈ SymkeysB
elem(2, decrypt(KBS,M)) = A
B→
KAB := elem(1, decrypt(KBS,M))
RB ∈ newnonce
(X,B,M) ←↩ Network
(respond, A) ←↩ StateB
(new session key, A,KAB, RB) ↪→ StateB
(B,A, (encrypt(KAB, RB))) ↪→ Network

SH Verification Tool:

def_trans_pattern B step_4
(X,A,S,M,KBS,RB)
(X,B,M) ? Network,
[’respond’,A] ? B_State,
[A,’agent’] ? B_State,
[S,’server’] ? B_State,
(S,’sym’,KBS) ? B_Symkeys,
head(tail(decrypt(KBS,M))) = A,

/* KAB = head(decrypt(KBS,M)),*/
/* RB ? new_nonce,*/

RB << new_nonce,
(X,B,M) << Network,
[’respond’,A] << B_State,
[new_session_key,A,head(decrypt(KBS,M)),RB] >> B_State,
(B,A,encrypt(head(decrypt(KBS,M)),[RB])) >> Network;

The usage chosen for nonces in the tool notation is described in step 1. In
the pattern aboveRA is bound to the list[17] , the single element of the multiset
new nonce , then the tail ofRA (in this case the empty list) is put back into the
multiset and the head ofRA (in this case17) is used in the further lines of the
above pattern.

APA model (step 5a):
Variables: X,B,M,KAB, RB

(X,A,M) ∈ Network
(new session key, B,KAB) ∈ StateA
A→
RB := elem(1, decrypt(KAB,M))
(X,A,M) ←↩ Network
(new session key, B,KAB) ←↩ StateA
(session key, B,KAB) ↪→ StateA
(B, sym,KAB) ↪→ SymkeysA
(A,B, (encrypt(KAB, (RB − 1)))) ↪→ Network

SH Verification Tool:

def_trans_pattern A step_5a
(X,B,M,KAB)
(X,A,M) ? Network,
[’new_session_key’,B,KAB] ? A_State,

/* RB = head(decrypt(KAB,M)),*/
(X,A,M) << Network,
[’new_session_key’,B,KAB] << A_State,
[’session_key’,B,KAB] >> A_State,
(B,’sym’,KAB) >> A_Symkeys,
(A,B,encrypt(KAB,[head(decrypt(KAB,M))-1])) >> Network;

APA model (step 5b):
Variables: X,A,M,KAB, RB

(X,B,M) ∈ Network
(new session key, A,KAB, RB) ∈ StateB
elem(1, decrypt(KAB,M)) + 1 = RB
B→
(X,B,M) ←↩ Network
(new session key, A,KAB, RB) ←↩ StateB
(session key, A,KAB) ↪→ StateB
(A, sym,KAB) ↪→ SymkeysB

SH Verification Tool:

def_trans_pattern B step_5b
(X,A,M,KAB,RB)
(X,B,M) ? Network,
[’new_session_key’,A,KAB,RB] ? B_State,
head(decrypt(KAB,M))+1 = RB,
(X,B,M) << Network,
[’new_session_key’,A,KAB,RB] << B_State,
[’session_key’,A,KAB] >> B_State,
(A,’sym’,KAB) >> B_Symkeys;

APA model (newA):
Variables: B,KAB

(session key, B,KAB) ∈ StateA
A→
(start, B) ↪→ StateA

SH Verification Tool:

def_trans_pattern A newA
(B,KAB)
[’session_key’,B,KAB] ? A_State,
[’start’,B] ˜? A_State,
[’start’,B] >> A_State;

In the tool notation it is checked whether the element[’start’,B] is already
in A State to prevent an endless loop here.

APA model (newB):
Variables: A,KAB

(session key, A,KAB) ∈ StateB
B→
(respond, A) ↪→ StateB

SH Verification Tool:

def_trans_pattern B newB
(A,KAB)
[’session_key’,A,KAB] ? B_State,
[’respond’,A] ˜? B_State,
[’respond’,A] >> B_State;

In the tool notation it is checked whether the element[’respond’,A] is
already inB State to prevent an endless loop here.

SH Verification Tool: Reachability graph
The reachability graph computed by the SH Verification Tool for this configu-

ration is shown in figures 3 and 4.

M-9

M-11

M-7M-8

M-10

M-6

M-4

M-5

M-3

M-2

M-1
start:

Alice_step_1

Server_step_2

Alice_step_3

Bob_step_4

Alice_step_5a

Alice_newABob_step_5b

Bob_step_5bAlice_newABob_newB

Bob_newBAlice_newA

Figure 3: Reachability graph

 Network = (Alice,Server,[Alice,Bob,13]) ) 
 Server_State = [Alice,agent].[Bob,agent] 
 Server_Symkeys = (Alice,sym,(Alice,Server,sym)).(Bob,sym,(Bob,Server,sym)) 
 M = [Alice,Bob,13] 
 X = Alice 
 A = Alice 
 B = Bob 
 KAS = (Alice,Server,sym) 
 KBS = (Bob,Server,sym) 
Server_step_2 (

 Network = (Server,Alice,[encrypt,(Alice,Server,sym),[13,Bob,(Alice,Bob,sym),[encrypt,(Bob,Server,sym),[(Alice,Bob,sym),Alice]]]]) ) 
 Alice_Symkeys = (Server,sym,(Alice,Server,sym)) 
 Alice_State = [Bob,13,Server].[Bob,agent].[Server,server] 
 M = [encrypt,(Alice,Server,sym),[13,Bob,(Alice,Bob,sym),[encrypt,(Bob,Server,sym),[(Alice,Bob,sym),Alice]]]] 
 X = Server 
 KAS = (Alice,Server,sym) 
 S = Server 
 RA = 13 
 B = Bob 
Alice_step_3 (

 Network = (Alice,Bob,[encrypt,(Bob,Server,sym),[(Alice,Bob,sym),Alice]]) ) 
 Bob_State = [Alice,agent].[Server,server].[respond,Alice] 
 Bob_Symkeys = (Server,sym,(Bob,Server,sym)) 
 new_nonce = [17] 
 M = [encrypt,(Bob,Server,sym),[(Alice,Bob,sym),Alice]] 
 X = Alice 
 A = Alice 
 S = Server 
 KBS = (Bob,Server,sym) 
 RB = [17] 
Bob_step_4 (

 Network = (Bob,Alice,[encrypt,(Alice,Bob,sym),[17]]) ) 
 Alice_State = [Bob,agent].[Server,server].[new_session_key,Bob,(Alice,Bob,sym)] 
 Alice_Symkeys = (Server,sym,(Alice,Server,sym)) 
 M = [encrypt,(Alice,Bob,sym),[17]] 
 X = Bob 
 KAB = (Alice,Bob,sym) 
 B = Bob 
Alice_step_5a (

 Alice_State = [Bob,agent].[Server,server].[session_key,Bob,(Alice,Bob,sym)] ) 
 KAB = (Alice,Bob,sym) 
 B = Bob 
Alice_newA (

 Network = (Alice,Bob,[encrypt,(Alice,Bob,sym),[16]]) ) 
 Bob_State = [Alice,agent].[Server,server].[new_session_key,Alice,(Alice,Bob,sym),17] 
 Bob_Symkeys = (Server,sym,(Bob,Server,sym)) 
 M = [encrypt,(Alice,Bob,sym),[16]] 
 X = Alice 
 RB = 17 
 KAB = (Alice,Bob,sym) 
 A = Alice 
Bob_step_5b (

 Network = (Alice,Bob,[encrypt,(Alice,Bob,sym),[16]]) ) 
 Bob_State = [Alice,agent].[Server,server].[new_session_key,Alice,(Alice,Bob,sym),17] 
 Bob_Symkeys = (Server,sym,(Bob,Server,sym)) 
 M = [encrypt,(Alice,Bob,sym),[16]] 
 X = Alice 
 RB = 17 
 KAB = (Alice,Bob,sym) 
 A = Alice 
Bob_step_5b (

 Alice_State = [Bob,agent].[Server,server].[session_key,Bob,(Alice,Bob,sym)] ) 
 KAB = (Alice,Bob,sym) 
 B = Bob 
Alice_newA (

 Bob_State = [Alice,agent].[Server,server].[session_key,Alice,(Alice,Bob,sym)] ) 
 KAB = (Alice,Bob,sym) 
 A = Alice 
Bob_newB (

 Bob_State = [Alice,agent].[Server,server].[session_key,Alice,(Alice,Bob,sym)] ) 
 KAB = (Alice,Bob,sym) 
 A = Alice 
Bob_newB (

 Alice_State = [Bob,agent].[Server,server].[session_key,Bob,(Alice,Bob,sym)] ) 
 KAB = (Alice,Bob,sym) 
 B = Bob 
Alice_newA (

Figure 4: Reachability graph with interpretation

Appendix

SH Verification Tool:

defset Agents = ’Alice’, ’Alice2’, ’Bob’, ’Server’ ;

defset Address = Agents;

defset Identity = Agents || nat_0;

defset Nonce = 13, 17, 19, 23;

defset Nonces = [Nonce];

defset Nonce_seq = seq(Nonces);

defset Run_seq = seq(nat_0);

defset Constants = ’start’, ’respond’, ’agent’, ’server’,
’new_session_key’, ’session_key’;

defset Symflags = ’sym’, ’symmac’;

defset Asymflags = ’pub’, ’priv’, ’pucipher’, ’privcipher’,
’pubverify’, ’privsign’;

defset Cryptfuncs = ’encrypt’, ’decrypt’, ’crypt’, ’sign’, ’mac’, ’hash’;

defset Keywords = Symflags || Asymflags || Constants || Cryptfuncs;

defset Skeys = pro(Agents, Agents, Symflags);

defset Akeys = pro(Agents, Asymflags);

defset Keys = Skeys || Akeys;

defset message = Keywords || Agents || Nonce || Keys || nat_0;
/* nat_0 wg. nonce -1 */

defset Messages = [message];

defset Messages_seq = seq(Messages);

defset Symkeys = pro(Identity, Symflags, Keys);

defset Asymkeys = pro(Identity, Asymflags, Keys);

defset Symkeys_seq = seq(Symkeys);

defset Asymkeys_seq = seq(Asymkeys);

defset net_elem = pro(Address, Address, Messages);

defset net_elem_seq = seq(net_elem);

defset nat0_seq = seq(nat_0);

def_state Network: net_elem_seq := ::;

def_state new_nonce: Nonce_seq := [13,17];
/*
def_state new_nonce: Nonce_seq := [13,17,19,23];
*/
/*
def_state new_runs: Run_seq := 1;
*/
def_role A;

def_role B;

def_role S;

def_state A State: Messages_seq := [B,’agent’].[S,’server’].[’start’,B];

def_state A Symkeys: Symkeys_seq := (S,’sym’,(A,S,’sym’));

def_state A Asymkeys: Asymkeys_seq := ::;

def_state B State: Messages_seq := [A,’agent’].[S,’server’].[’respond’,A];

def_state B Symkeys: Symkeys_seq := (S,’sym’,(B,S,’sym’));

def_state B Asymkeys: Asymkeys_seq := ::;

def_state S State: Messages_seq := [A,’agent’].[B,’agent’];

def_state S Symkeys: Symkeys_seq := (A,’sym’,(A,S,’sym’)).
(B,’sym’,(B,S,’sym’));

def_state S Asymkeys: Asymkeys_seq := ::;

/*
defcase encrypt: pro(Keys,Messages) >> Messages

encrypt(key,message) = if head(message) = ’decrypt’ &
head(tail(message)) = key

then head(tail(tail(message)))
else [’encrypt’,key,message];

*/

defterm encrypt: pro(Keys,Messages) >> Messages
encrypt(key,message) = [’encrypt’,key,message];

defcase decrypt: pro(Keys,Messages) >> Messages
decrypt(key,message) = if head(message) = ’encrypt’ &

head(tail(message)) = key
then head(tail(tail(message)))
else [’decrypt’,key,message];

def_trans_pattern A step_1
(RA,B,S)
[’start’,B] ? A_State,
[B,’agent’] ? A_State,
[S,’server’] ? A_State,
RA << new_nonce,
tail(RA) >> new_nonce,
[’start’,B] << A_State,
[B,head(RA),S] >> A_State,
(A,S,[A,B,head(RA)]) >> Network;

def_trans_pattern S step_2
(X,A,B,M,KAS,KBS)
(X,S,M) ? Network,
[A,’agent’] ? S_State,
A = head(M),
[B,’agent’] ? S_State,
B = head(tail(M)),
(A,’sym’,KAS) ? S_Symkeys,
(B,’sym’,KBS) ? S_Symkeys,

/* RA = head(tail(tail(M))),*/
(X,S,M) << Network,
(S,A,encrypt(KAS,[head(tail(tail(M))),B,(A,B,’sym’),

encrypt(KBS,[(A,B,’sym’),A])])) >> Network;

def_trans_pattern A step_3
(X,B,S,M,KAS,RA)
(X,A,M) ? Network,

/* [B,RA,S] ? A_State,*/
(S,’sym’,KAS) ? A_Symkeys,
head(decrypt(KAS,M)) = RA,
head(tail(decrypt(KAS,M))) = B,
[B,RA,S] ? A_State,

/* KAB = head(tail(tail(decrypt(KAS,M)))),*/
/* Chiffretext = head(tail(tail(tail(decrypt(KAS,M))))),*/

(X,A,M) << Network,
[B,RA,S] << A_State,
[new_session_key,B,head(tail(tail(decrypt(KAS,M))))] >> A_State,
(A,B,head(tail(tail(tail(decrypt(KAS,M)))))) >> Network;

def_trans_pattern B step_4
(X,A,S,M,KBS,RB)
(X,B,M) ? Network,
[’respond’,A] ? B_State,
[A,’agent’] ? B_State,
[S,’server’] ? B_State,
(S,’sym’,KBS) ? B_Symkeys,
head(tail(decrypt(KBS,M))) = A,

/* KAB = head(decrypt(KBS,M)),*/
/* RB ? new_nonce,*/

RB << new_nonce,
tail(RB) >> new_nonce,
(X,B,M) << Network,
[’respond’,A] << B_State,
[new_session_key,A,head(decrypt(KBS,M)),head(RB)] >> B_State,
(B,A,encrypt(head(decrypt(KBS,M)),[head(RB)])) >> Network;

def_trans_pattern A step_5a
(X,B,M,KAB)
(X,A,M) ? Network,
[’new_session_key’,B,KAB] ? A_State,

/* RB = head(decrypt(KAB,M)),*/
(X,A,M) << Network,
[’new_session_key’,B,KAB] << A_State,
[’session_key’,B,KAB] >> A_State,
(B,’sym’,KAB) >> A_Symkeys,
(A,B,encrypt(KAB,[head(decrypt(KAB,M))-1])) >> Network;

def_trans_pattern B step_5b
(X,A,M,KAB,RB)
(X,B,M) ? Network,
[’new_session_key’,A,KAB,RB] ? B_State,
head(decrypt(KAB,M))+1 = RB,

(X,B,M) << Network,
[’new_session_key’,A,KAB,RB] << B_State,
[’session_key’,A,KAB] >> B_State,
(A,’sym’,KAB) >> B_Symkeys;

def_trans_pattern A newA
(B,KAB)
[’session_key’,B,KAB] ? A_State,
[’start’,B] ˜? A_State,
[’start’,B] >> A_State;

def_trans_pattern B newB
(A,KAB)
[’session_key’,A,KAB] ? B_State,
[’respond’,A] ˜? B_State,
[’respond’,A] >> B_State;

def_pattern_bind A := ’Alice’ ;

def_pattern_bind B := ’Bob’ ;
/*
def_pattern_bind B := ’Alice2’, ’Bob’ ;
*/
def_pattern_bind S := ’Server’ ;

